

Configurare automaticamente l'adattamento a schermo di WindowViewer per tutti gli utenti

Introduzione

Questa TN descrive le operazioni da seguire per adattare dinamicamente la visualizzazione dell'applicazione alla risoluzione corrente dello schermo.

Versioni


Quanto descritto in questa TN è stato verificato sulle seguenti versioni:

- Versioni dalla 2017 U3 SP1 in avanti

Descrizione del problema

Quando si desidera che WindowViewer adatti dinamicamente la visualizzazione dell'applicazione alla risoluzione corrente dello schermo, è necessario assicurarsi che il file di configurazione Win.ini contenga la chiave `ViewApplicationResolution=2`.

Tale chiave viene scritta da Application Manager quando si va ad agire nel menu *Tools > Node Properties > Resolution*:



Relativamente a tale menu, i possibili valori della chiave `ViewApplicationResolution` sono i seguenti:

- 0 Checkbox "Allow WindowViewer to dynamically change resolution" deselezionata (**default**)
- 1 Opzione "Use application resolution"
- 2 Opzione "Convert to screen video resolution" (valore desiderato nel caso quiesposto)
- 3 Opzione "Custom resolution"

Poiché ogni utente di Windows ha il proprio file Win.ini nel percorso:

`C:\Users\<username>\AppData\Local\Wonderware\Win.ini`

modificare la configurazione può diventare complicato in ambienti multi-user, come ad esempio Terminal Server acceduti tramite connessione Remote Desktop standard di Windows oppure tramite **InTouch Access Anywhere**, soprattutto quando l'applicazione viene aperta proprio al primo login dell'utente in Windows, quindi non vi possono essere occasioni precedenti per poter modificare il relativo file Win.ini, in quanto la directory `C:\Users\<username>` non esisteva e viene creata contestualmente all'apertura della sessione di supervisione (situazione molto frequente).

E' necessario inoltre inserire anche la chiave **ScaleForResolution**.

Il valore del parametro ScaleForResolution determina se le finestre dell'applicazione (*.win) vengono ridimensionate automaticamente da WindowMaker dopo la modifica della risoluzione dello schermo sul computer che esegue WindowViewer.

Il parametro **ScaleForResolution** non influisce sulla risoluzione delle finestre di dialogo WindowViewer.

Per configurare la conversione della risoluzione dello schermo su un nodo

- Modificare il file **InTouch.ini** del computer che esegue InTouch.
- Aggiungere il parametro **ScaleForResolution=1** al file.

Se impostato su 0, la conversione della risoluzione è disabilitata.

Se impostato su 1, la conversione della risoluzione è abilitata.

Nel caso di un'InTouchView Application, il file è situato in:

`"C:\Program Files (x86)\ArchestraA\Framework\Bin\<GalaxyName>-<NomeIstanzaInTouch>\"`

Nota: se il parametro ScaleForResolution non viene aggiunto al file InTouch.ini, il valore predefinito è abilitato (ScaleForResolution=1). Quando si disabilita il parametro (ScaleForResolution=0), viene comunque richiesto di convertire la risoluzione. Tuttavia, la conversione della risoluzione non avviene.

Configurazione automatica mediante script di InTouch

È possibile fare in modo che, subito dopo il login dell'operatore all'interno dell'applicazione, uno script vada a controllare se il file Win.ini dell'utente è da correggere, quindi faccia eventualmente ripartire WindowMaker per rendere effettiva la modifica della configurazione.

Di seguito è proposto un esempio di codice per scripting InTouch classico (utilizzabile quindi in qualsiasi tipo di applicazione), da richiamare su evento DataChange della tag di sistema **\$Operator**. Lo script prevede l'esistenza di una tag di tipo *Memory Message* chiamata **Line**.

```
IF $Operator <> "None" THEN
  DIM path AS MESSAGE;
  path = "C:\Users\" + $Operator + "\AppData\Local\Wonderware\Win.ini";

  DIM offset_start AS INTEGER;
  DIM offset_end AS INTEGER;
  DIM i AS INTEGER;

  FOR i = 1 TO 10000
    {Leggo la prossima riga e memorizzo il byte offset di fine lettura}
    offset_end = FileReadMessage(path, offset_start, Line, 0);

    {
      Se il file è terminato, oppure contiene già la chiave ViewApplicationResolution,
      mi fermo senza aggiornare il byte offset di inizio lettura
      (resta quello dell'ultima riga letta)
    }

    IF offset_end == 0 OR StringLeft(Line, 25) == "ViewApplicationResolution" THEN
      EXIT FOR;
    ENDIF;

    {Non mi sono fermato; aggiorno il byte offset di inizio lettura
    in modo che la prossima read parta da dove è finita la precedente}
    offset_start = offset_end;
  NEXT;

  {
    Se avevo letto tutto il file senza trovare la chiave ViewApplicationResolution,
    il byte offset di inizio riga punta alla fine del file Win.ini, quindi la write
    va ad ACCODARE al file la nuova riga "ViewApplicationResolution=2";
    Altrimenti, controllo se la chiave aveva valore diverso da 2, nel qual caso la write
    va a SOVRASCRIVERE l'intera riga con "ViewApplicationResolution=2".
    In entrambi i casi, riavvio WindowViewer per rendere effettiva la modifica
  }

  IF offset_end == 0 OR StringRight(Line, 1) <> "2" THEN
    FileWriteMessage(path, offset_start, "ViewApplicationResolution=2", 0);
    RestartWindowViewer();
  ENDIF;
ENDIF;
```

Si noti che tutte le funzioni utilizzate sono native, quindi non sono necessarie DLL aggiuntive. Con alcuni adattamenti, potendo usare metodi del framework .NET oppure funzioni non native quali ad esempio InfoDosEnv(), è possibile evitare la necessità di login esplicito da parte dell'operatore prima dell'eventuale riconfigurazione del file Win.ini e del conseguente riavvio di WindowViewer.

Tali metodi non verranno approfonditi nel presente documento, che vuole esclusivamente suggerire metodologie di principio, da cui sia possibile derivare le personalizzazioni necessarie.

Attenzione: lo script proposto è puramente indicativo e funziona correttamente solo per i file Win.ini ben formati, nonché privi di spazi aggiuntivi ad inizio/fine riga o a cavallo del carattere '='. La funzionalità qui esposta può essere implementata in numerosi modi differenti, e il supporto tecnico Wonderware Italia non è responsabile di eventuali malfunzionamenti derivanti dall'utilizzo del codice suddetto all'interno di qualsivoglia progetto specifico. Così come presentato, l'esempio associa direttamente la tag \$Operator al nome della directory assegnata all'utente di Windows, perciò **il codice tal quale non è generalmente applicabile a progetti configurati con Security type InTouch o ArchestrA / Galaxy.**

Disclaimer

Il presente documento è fornito a scopo di esempio e non sostituisce la documentazione AVEVA. L'applicazione di quanto contenuto, in un preciso ambito applicativo, deve essere sempre validata da un tecnico Wonderware. La documentazione rilasciata da AVEVA resta il riferimento tecnico ufficiale da seguire: softwaresupport.aveva.com. Wonderware Italia non si assume la responsabilità di un'applicazione scorretta di questo documento.